

VHDL Presentation

Alex Boswell

What is a Hardware Description Language?

The short and sweet:

- Hardware Description Languages are a textual description of a digital circuit
- Hardware Description Languages are also used to simulate a digital circuit
- Hardware Description Languages borrow syntax and style from procedural programming languages (C, C++, Ada).
- Programming languages are just high level abstractions of instructions for a CPU to execute.
- Hardware Description Languages can ultimately be used to design a CPU
- Traditionally HDLs were used to describe the behavior of Application Specific Integrated Circuits (ASICs) and provide simulations for said Integrated Circuits.
- Since the mid 1990s, Hardware Description Languages can use synthesis software to create a digital circuit inside of a programmable logic device (PLD, CPLD, FPGA).
- The two dominant Hardware Description languages are SystemVerilog and VHDL.

Basic Structure of a VHDL file

A typical VHDL design file has the following regions

- library declaration
- package imports
- entity declaration
- architecture declarative region
- architecture behavioral region

```
1  library libraryname; --library declare
2  use libraryname.packagename.all; --package import
3
4  entity entity_name is --entity declaration
5      port(
6          identifier : mode type;
7          identifier : mode type
8      );
9  end entity;
10
11 architecture arch_name of entity_name is
12     signal identifier : type; --architecture declarative region
13     constant identifier : type := initialvalue;
14     function name is blah blah blah;
15     procedure name is blah blah blah;
16     type type_name is (definition);
17 begin
18     behavioral stuff --architecture behavioral region
19 end architecture;
```

Describe the ports, signals, constants, functions, procedures, types to the class.

Common packages:

- std_logic_1164
- numeric_std
- math_real

Basic Operators

Inside of the architecture behavioral region you can assign logic and/or values to signals and ports using `<=`.

The basic operators in VHDL are **AND**, **OR**, **XOR**, **NAND**, **NOR**, **XNOR**, **NOT**.

Given a VHDL file with input ports a, b and output port f :

```
1 f <= a xor b;
```

Because this is not a programming language, two things must be made clear:

1. Given multiple outputs, the order of assignment does not matter since they represent individual gates
2. Because this order does not matter, you cannot reassign a signal or port. VHDL sees this as a conflicting driver rather than a reassignment.

When it comes to the order of operations, there is no set precedent, except for **NOT** (this always goes first, so **ALWAYS** use parenthesis to denote the logic grouping).

```
1 f <= ((not a and b) or (c nand d)) xor e;
```

This is the equivalent of

$$((\overline{A} \cdot B) + (\overline{C \cdot D})) \oplus E$$

The Fundamental Type

While there is a built-in type **bit**, the most used type due to its versatility is **std_logic**. This type is imported with the **ieee.std_logic_1164** package.

While bit has two values, either '1' or '0', std_logic has a total of nine values for superior simulation capabilities. Of those nine, only four can be used for synthesis.

- '1' A logic level high (Synthesizable)
- '0' A logic level low (Synthesizable)
- 'Z' A high impedance output (Synthesizable)
- '-' A don't care input (Synthesizable)
- 'U' Uninitialized
- 'X' Unknown
- 'W' Weak Unknown
- 'L' Weak logic level low
- 'H' Weak logic level high

The '1' and '0' are, hopefully, self explanatory in what they do. The 'Z' operator forces an output to be the high impedance value. Neither a HIGH or a LOW. These are commonly found in tri-state buffers. Tri-state buffers are mainly used in busses that connect multiple devices, in order to ensure only one device is transmitting on the bus.




Figure 1: Tri-state buffer and truth table

The '-' is used to simplify repeated inputs and will be explored later.

Standard Logic Vector

The `std_logic_vector` is an array of the type `std_logic`. Unlike traditional programming languages, you must define the constraints of the array. The overwhelming majority of the time you want to have the vector is the big endian format, thus you declare it as a `downto`.

```
1 a: in std_logic_vector(7 downto 0); --a(7)...a(0)
2 b: in std_logic_vector(2 to 5); --b(2)...b(5)
```

The way operators work on vectors is different than that of scalar values. If two vectors of equal length are operated on, the result is a vector of the same size in which each element is operated with the corresponding element. If a vector is operated on by a scalar, the result is a vector of equal length in which each element has been operated on by the scalar. If an operator is before the vector, every element in the vector is operated together, making a scalar. This is a mouthful, so here is an example:

```
1 signal a: std_logic_vector(3 downto 0) := "1011";
2 signal b: std_logic_vector(3 downto 0) := "1101";
3 signal c: std_logic := '1';
4 signal f: std_logic_vector(3 downto 0);
5 signal g: std_logic_vector(3 downto 0);
6 signal h: std_logic;
7 f <= a and b; -- "1001";
8 g <= a xor c; -- "0100";
9 h <= and b; -- b(3) and b(2) and b(1) and b(0) = '0'
```

As can be seen in the comments for line 9, given a vector "b", you can index vector b with b(i) where i is an integer within the specified range of the vector.

Two ways to join elements into a vector are aggregation and concatenation. For aggregation, you can define specific indexes of a vector to be something, and then mass assign the remainder to be whatever. This is particularly helpful if you have a massive vector. Concatenation is mainly used to join vectors together.

```
1 signal s1: std_logic_vector(3 downto 0) := "1011";
2 signal s2: std_logic_vector(3 downto 0) := "1101";
3 signal b1: std_logic_vector(7 downto 0);
4 signal ag1: std_logic_vector(3 downto 0);
5 signal ag2: std_logic_vector(7 downto 0);
6 b1 <= s1 & s2; -- "10111101"
7 ag1 <= (3|1 => '1', others => '0'); -- "1010"
8 ag2 <= (7 downto 5 => 'Z', 4|2|0 => '-', others => s2(2));
9 -- "ZZZ-1-1"
```

Signals as wires

Signals are constructs in VHDL that can act as either wires or registers. Right now, only the wire usage shall be explored. The declaration for a signal occurs in the architecture declarative body.

```
1 architecture rtl of entity_name is
2     signal identifier : std_logic := '0';
3 begin
4
5 end architecture;
```

Here is an example of a Full Adder using signals as wires and the schematic equivalent.

```
1 library ieee;
2 use ieee.std_logic_1164.all;
3
4 entity fullAdder is
5     port(
6         a,b,c: in std_logic; --input/output declaration
7         sum,carry: out std_logic
8     );
9 end fullAdder;
10
11 architecture behavior of fullAdder is
12     signal w1,w2,w3: std_logic; --intermediate variables
13 begin
14
15     w1 <= a xor b; --logic
16     w2 <= w1 and c;
17     w3 <= a and b;
18     sum <= w1 xor c;
19     carry <= w2 or w3;
20
21 end behavior;
```




Figure 2: Full Adder

Select Statement

Select is the first of the concurrent statements we will go over. Select is analogous to a switch statement in C++. The main purpose of this is to make truth tables and should be the primary usage of this. The main Syntax for it is:

```
1 with expression select
2     target <= value when choice,
3         value when choice,
4         value when choice,
5         value when others;
6 -- when others is recommended for simulation purposes.
```

Example Multiplexer:

```
1 library ieee;
2 use ieee.std_logic_1164.all;
3
4 entity mux is
5     port(
6         D      : in  std_logic_vector(3 downto 0);
7         sel     : in  std_logic_vector(1 downto 0);
8         mux_out : out std_logic
9     );
10 end mux;
11
12 architecture rtl of mux is
13
14 begin
15     with SEL select
16         mux_out <= D(3) when "11",
17                 D(2) when "10",
18                 D(1) when "01",
19                 D(0) when "00",
20                 '0' when others;
21 end architecture;
```


When Statement

The when statement is analogous to an if, else if, else construct. As such, it has more freedom for the equality matching, just like an if statement vs a switch in a real programming language.

Syntax:

```
1 target <= value when condition1 else
2     value when condition2 else
3     value when condition3 else
4     value;
```

What makes it more powerful than the select statement is that it can use equality operators and any signal to achieve the desired output for the target.

Here is an example of a comparator:

```
1 library ieee;
2 use ieee.std_logic_1164.all;
3
4 entity comparator is
5     generic(
6         n : integer := 8
7     );
8     port(
9         a, b : in std_logic_vector(n-1 downto 0);
10        eq, neq, lt, lte, gt, gte : out std_logic
11    );
12 end;
13
14 architecture synth of comparator is
15
16 begin
17     eq <= '1' when (a = b) else '0'; --equal to
18     neq <= '1' when (a /= b) else '0'; --not equal to
19     lt <= '1' when (a < b) else '0'; --less than
20     lte <= '1' when (a <= b) else '0'; --less than equal to
21     gt <= '1' when (a > b) else '0'; --greater than
22     gte <= '1' when (a >= b) else '0'; --greater than equal to
23 end architecture;
```

With logical operators like **AND** and **OR**, you can add significantly more conditions to be tested.

When statements have an inherent priority to them since they act like if, else if statements, so one can create priority encoders using the when statement as well. **The select statement does not have this property.** Here is an 8-3 line encoder.

```
1 outdata <= "111" when data(7) = '1' else
2     "110" when data(6) = '1' else
3     "101" when data(5) = '1' else
4     "100" when data(4) = '1' else
5     "011" when data(3) = '1' else
6     "010" when data(2) = '1' else
7     "001" when data(1) = '1' else
8     "000" when data(0) = '1' else
9     "000";
```

Generate statement

The generate statement is a shortcut to duplicate combinational assignments. It works by printing the statements generated by the loop multiple times. A piece of hardware is inferred with every loop, hence why generate was chosen as the name

Syntax:

```
1  label:for i in range generate
2      --statements
3  end generate;
```

Here are a few examples:

```
1  --reverse a vector
2  reverse: for i in a'range generate
3      b(i) <= a(a'high-i);
4  end generate;
5  --xor LSB of a vector with MSB of another vector
6  gen: for i in x'range generate
7      x(i) <= a(i) xor b(b'left-i);
8  end generate;
9  --3-8 line decoder
10 entity combogen is
11     port(
12         input: in std_logic_vector(2 downto 0);
13         output: out std_logic_vector(7 downto 0)
14     );
15 end combogen;
16
17 architecture rtl of combogen is
18     signal counter: integer := 0;
19 begin
20     gen: for i in 0 to 7 generate
21         output(i) <= '1' when counter = i else '0';
22         end generate;
23     counter <= to_integer(unsigned(input));
24 end architecture;
```

Functions and attributes

As you have have noticed in the previous code block, there are two concepts not covered. The 'left, 'high, 'range and the function to_integer.

The one with the ticks are called attributes. Attributes can access information about the properties of either a signal, type, or entity.

For example, 'range is short hand for specifying the range of a vector (0 to 4), (5 downto 3), whatever.

Here are a list of some noteworthy attributes in VHDL: T means type, A means array, S means scalar.

```
1  attribute ascending      : boolean; -- T'ASCENDING is boolean true if range of T defined with to .
2                                -- A'ASCENDING is boolean true if range of A defined with to .
3  attribute event          : boolean; -- S'EVENT is true if signal S has had an event this simulation cycle.
4  attribute high           : integer; -- T'HIGH is the highest value of type T.
5                                -- A'HIGH
6  attribute left           : integer; -- T'LEFT is the leftmost value of type T. (Largest if downto)
7                                -- A'LEFT
8  attribute length         : integer; -- A'LENGTH is the integer value of the number of elements in array A.
9  attribute low            : integer; -- T'LOW is the lowest value of type T.
10                               -- A'LOW
11 attribute range          : string;  -- A'RANGE is the range A'LEFT to A'RIGHT or A'LEFT downto A'RIGHT .
12 attribute reverse_range  : string;  -- A'REVERSE_RANGE is the range of A with to and downto reversed.
13 attribute stable         : string;  -- S'STABLE is true if no event is occurring on signal S.
14                               -- S'STABLE(t) is true if no even has occurred on signal S for t time.
```

The function to_integer is defined in the package numeric_std. It takes an signed or unsigned binary input and converts it to its respective integer representation. This is why we see input being casted to an unsigned by doing *unsigned(input)*.

If you want to convert an integer to unsigned binary and then to a std_logic_vector, you would do:

std_logic_vector(to_unsigned(int,n)), where n is the bit length of the vector.

There are a bunch of useful functions in the various packages for you to explore.

Sequential Logic: The process

A process in VHDL is an extremely difficult concept to understand for someone coming from a programming background. The typical syntax of a process is:

```
1 architecture
2 begin
3     process(sensitivity_list)
4         --declarative region
5     begin
6         --statements
7     end process;
8 end architecture;
```

If you research online, you may hear that the process is a region in which you can do sequential statements. This is misleading. While it is true you can do sequential constructs, signal assignments are still concurrent. You CANNOT do a loop 5 times and expect a integer signal to go up by 5 in a single process reevaluation.

Every time a signal in the sensitivity list experiences a change, or an event, the process is reevaluated. Typically when we do sequential statements, the process is only sensitive to the clock. But we never want to imply a latch, so we use a nifty function called *rising_edge()* in order for the language to imply an edge-triggered flip-flop rather than a latch. Here is an example of a D Flip-Flop:

```
1 architecture rtl of combogen is
2     signal clk,d,q,qb : std_logic;
3 begin
4     process(clk)
5     begin
6         if rising_edge(clk) then
7             q <= d;
8             qb <= not d;
9         end if;
10    end process;
11 end architecture;
```

Why not have qb be not q? Well, the way assignment happens with processes is that at the very end of the processes, the signal assignments are applied all at the same time. Meaning that even though q is now d, qb is going to be whatever q was last event, not what it is now. PULL UP THE DIGITAL SIMULATION AS A REFERENCE.

The if statement

The if statement is a processed version of the when statement, though as of VHDL-2008 whens can be used in processes. But we don't care right now.

An example can of course be seen on the previous page. Syntax:

```
1  if condition1 = 4 or condition2 /= "hello" then
2      --statements
3  elsif condition3 < 27 and condition3 /= "meow" then
4      ---statements
5  else
6      report "I love VHDL so much :)))";
7  end if;
```

The case statement

The case statement is analogous to the the select statement in conditional space. However, case statements are capable of nesting if statements within them. An example of this can be seen in the skeleton code. The generic syntax is:

```
1  case signal is
2      when (state) =>
3          -- statements
4      when (state2) | (state3) => -- either state2 or state3
5          if condition then --thing
6          else -- thing
7          end if;
8      when others =>
9          null --do nothing
10 end case;
```

Case statements are the best choice for doing state machines due to their flexible control structure.

The loop statement

The loop statement is practically identical to the generate statement (it does not require a label), however it gains two keywords for better control flow over the loop. **exit** and **next**.

During loops, using the exit statement, of course, exits the loop. This is useful if you want to make a gigantic if elsif construct without having to write an absurd amount of lines. Next is mainly used to make patterns, such as excluding even numbers during a loop.

```
1  entity priorityencode is
2      port(
3          a1: in std_logic_vector(7 downto 0);
4          b1: out std_logic_vector(2 downto 0)
5      );
6  end priorityencode;
7
8  architecture rtl of priorityencode is
9      function priority_encode(input: std_logic_vector) return std_logic_vector is
10         variable result: std_logic_vector(integer(ceil(log2(real(input'length)))) - 1 downto 0) := (others => '0');
11     begin
12         for i in input'range loop
13             if input(i) = '1' then
14                 result := std_logic_vector(to_unsigned(i,result'length));
15                 exit;
16             end if;
17         end loop;
18         return result;
19     end function priority_encode;
20
21 begin
22     b1 <= priority_encode(a1);
23 end architecture;
```

```
1  process(clk, reset)
2      begin
3          if reset = '1' then
4              reg <= (others => '0');
5          elsif rising_edge(clk) then
6              loadt <= load;
7              rbtnt <= rbtn;
8              lbtnt <= lbtn;
9              if load = '1' and loadt = '0' then
10                 reg <= swts;
11             elsif lbtn = '1' and lbtnt = '0' then
12                 reg <= reg(6 downto 0) & '0';
13             elsif rbtn = '1' and rbtnt = '0' then
14                 for i in 0 to 7 loop --EXAMPLE OF NEXT STATEMENT!
15                     if i mod 2 = 0 then --switch 0 to 1 if you want odds
16                         next; --If the index is divisible by 2, skip the loop.
17                     end if;
18                     reg(i) <= '0';
19                 end loop;
20             end if;
21         end if;
22     end process;
```

Counters

Here is an example of a 4 bit counter with down up capabilities:

```
1  library ieee;
2  use ieee.std_logic_1164.all;
3  use ieee.numeric_std.all;
4
5  entity counter is
6      port(
7          clk : in  std_logic;
8          du  : in  std_logic;
9          q   : out std_logic_vector(3 downto 0)
10     );
11 end counter;
12
13 architecture rtl of counter is
14     signal count : unsigned(3 downto 0) := (others => '0');
15 begin
16     process(clk)
17     begin
18         if rising_edge(clk) then
19             if du = '1' then
20                 count <= count - 1;
21             elsif du = '0' then
22                 count <= count + 1;
23             end if;
24         end if;
25     end process;
26     q <= std_logic_vector(count);
27 end architecture;
```

Another way of doing it:

```
1  library ieee;
2  use ieee.std_logic_1164.all;
3  use ieee.numeric_std.all;
4
5  entity counter is
6      port(
7          clk: in  std_logic;
8          du: in  std_logic;
9          q: out std_logic_vector(3 downto 0)
10     );
11 end counter;
12
13 architecture rtl of counter is
14     signal count: std_logic_vector(3 downto 0) := "0000";
15 begin
16     process(clk)
17     begin
18         if rising_edge(clk) then
19             if du = '0' then
20                 count <= std_logic_vector(unsigned(count) + 1);
21             elsif du = '1' then
22                 count <= std_logic_vector(unsigned(count) - 1);
23             end if;
24         end if;
25     end process;
26     q <= count;
27 end architecture;
```

Shift Registers

Shift registers can be made via the concatenation operator &.

```
1  library ieee;
2  use ieee.std_logic_1164.all;
3  use ieee.numeric_std.all;
4
5  entity SIPO is
6      port(
7          clk: in std_logic;
8          din: in std_logic;
9          rw: in std_logic;
10         dout: out std_logic_vector(7 downto 0)
11     );
12 end SIPO;
13
14 architecture rtl of SIPO is
15     signal datastore: std_logic_vector(7 downto 0) := (others => '0');
16 begin
17     process(clk)
18     begin
19         if rising_edge(clk) and rw = '0' then
20             datastore <= datastore(6 downto 0) & din;
21         end if;
22     end process;
23
24     dout <= datastore when rw = '1' else (others => 'Z');
25 end architecture;
```


Enumerated Types

One of the core prerequisites to making a state machine is using custom types.

VHDL is extremely versatile when it comes to types. This is because of its basis in the Ada programming language.

Enumerated types can be created with the following syntax.

```
1 architecture rtl of SIPO is
2     type enumerated_type is (first, second, third, fourth, fifth);
3     signal thing : enumerated_type := second;
4 begin
5
6 end architecture;
```

Enumerated types are primarily used in state machines rather than binary. The reason is that the synthesis tools will optimize for the flip flop allocation. You can also manually control it, and that will be explained later. For now let us continue to examples of state machines.

State Machines

State machines are just clocked processes with a case and conditional branching. An example of a state machine can be seen in lab 6. Here are some other random state machines.

A different version of lab 6 for FPGAs:

```
1  library ieee;
2  use ieee.std_logic_1164.all;
3  use ieee.numeric_std_unsigned.all;
4
5  entity mymachine is
6      port(
7          clk: in std_logic;
8          areset: in std_logic;
9          sreset: in std_logic;
10         change: in std_logic;
11         sseg: out std_logic_vector(6 downto 0)
12     );
13 end mymachine;
14
15 architecture rtl of mymachine is
16     type stateType is (B, A, D, E); --creates a new type
17     signal next_state, state: stateType := B; --assigns signals to type
18 begin
19     process(clk,areset)
20     begin
21         if areset = '1' then
22             state <= B;
23         elsif rising_edge(clk) then
24             if sreset = '1' and areset = '0' then
25                 state <= A;
26             else
27                 state <= next_state;
28             end if;
29         end if;
30     end process;
31
32     process(state,change)
33     begin
34         case state is
35             when B =>
36                 if change = '1' then
37                     next_state <= A;
38                 else
39                     next_state <= E;
40                 end if;
41             when A =>
42                 next_state <= D;
43             when D =>
44                 if change = '1' then
45                     next_state <= E;
46                 else
47                     next_state <= B;
48                 end if;
49             when E =>
50                 if change = '1' then
51                     next_state <= B;
52                 else
53                     next_state <= A;
54                 end if;
55             end case;
56         end process;
57
58         sseg <= "1100000" when state = B else
59                 "0110000" when state = E else
60                 "0001000" when state = A else
61                 "1000010" when state = D else "0000000";
62     end architecture;
```

As can be seen, the machine is split up into two processes. One for the clock and the other for selecting the next state. In FPGAs this is the most ideal method for describing state machines.

Manual encoding:

```
1  library ieee;
2  use ieee.std_logic_1164.all;
3
4  entity async is
5      port(
6          clk: in std_logic;
7          areset: in std_logic;
8          input: in std_logic;
9          output: out std_logic
10     );
11 end async;
12
13 architecture rtl of async is
14     constant A: std_logic_vector (1 downto 0) := "00";
15     constant B: std_logic_vector (1 downto 0) := "01";
16     constant C: std_logic_vector (1 downto 0) := "10";
17     constant D: std_logic_vector (1 downto 0) := "11";
18     signal state: std_logic_vector (1 downto 0);
19     signal next_state: std_logic_vector (1 downto 0);
20 begin
21     clockBehavior: process(clk,areset)
22     begin
23         if areset = '1' then
24             state <= A;
25         elsif rising_edge(clk) then
26             state <= next_state;
27         else
28             state <= state;
29         end if;
30     end process clockBehavior;
31
32     stateSwitching: process(state, input)
33     begin
34         case state is
35             when A =>
36                 if (input = '1') then
37                     next_state <= B;
38                 elsif (input = '0') then
39                     next_state <= A;
40                 end if;
41             when B =>
42                 if (input = '1') then
43                     next_state <= C;
44                 elsif (input = '0') then
45                     next_state <= B;
46                 end if;
47             when C =>
48                 if (input = '1') then
49                     next_state <= D;
50                 elsif (input = '0') then
51                     next_state <= C;
52                 end if;
53             when D =>
54                 if (input = '1') then
55                     next_state <= A;
56                 elsif (input = '0') then
57                     next_state <= D;
58                 end if;
59             when others =>
60                 next_state <= A;
61         end case;
62     end process stateSwitching;
63     output <= '1' when state = D else '0';
64 end architecture;
```

State machine theory is extensive. The best I have seen is in chapter 15 of the book:

Circuit Design with VHDL 3rd ed. by Volnei A. Pedroni

Synthesis Attributes

Creating a custom attribute for synthesis is easy. The syntax is:

```
1 attribute attribute_name : attribute_type;  
2 attribute attribute_name of entity_name : entity_class is expression;
```

Each synthesis tool has a wide breadth of synthesis attributes to work with. For the most part all we care about are

```
1 attribute LOC : string;  
2 attribute syn_encoding : string;
```

LOC is declared in the entity and the string value is "P##" where the numbers represent the pin numbers. For example "P01" would be pin 1.

syn_encoding can be one of the following:

- sequential
- gray
- onehot

This attribute tells the synthesizer what to do with the registers when an enumerated state type is used.

Example:

```
1 signal state : state_type := First;  
2 attribute syn_encoding of state : signal is "onehot";
```

```

1  library ieee;
2  use ieee.std_logic_1164.all;
3  use ieee.numeric_std.all;
4  use ieee.math_real.all;
5  use work.UART2.all;
6
7  entity BASYS_UART_RX is
8      port(
9          clock          : in  std_logic;
10         serial_input   : in  std_logic;
11         stop_bit       : out std_logic;
12         data_byte      : out std_logic_vector(UART_BITS-1 downto 0)
13     );
14 end entity;
15
16 architecture rtl of BASYS_UART_RX is
17     type UART_RX_SM is (RX_Idle, RX_Start, RX_Data, RX_Stop, RX_Cleanup);
18     signal rx_state      : UART_RX_SM := RX_Idle;
19     signal clock_count   : integer range 0 to UART_CLKB-1 := 0;
20     signal bit_index     : integer range 0 to UART_BITS-1 := 0;
21     signal register_byte : std_logic_vector(UART_BITS-1 downto 0) := (others => '0');
22     signal wire_stop_bit : std_logic := '0';
23 begin
24     statemachine:process(clock)
25     begin
26         if rising_edge(clock) then
27             case rx_state is
28                 when RX_Idle =>
29                     wire_stop_bit <= '0';
30                     clock_count <= 0;
31                     bit_index <= 0;
32
33                     if serial_input = '1' then
34                         rx_state <= RX_Start;
35                     else
36                         rx_state <= RX_Idle;
37                     end if;
38                 when RX_Start =>
39                     if clock_count = (UART_CLKB-1) / 2 then
40                         if serial_input = '0' then
41                             clock_count <= 0;
42                             rx_state <= RX_Data;
43                         else
44                             rx_state <= RX_Start;
45                         end if;
46                     else
47                         clock_count <= clock_count + 1;
48                         rx_state <= RX_Start;
49                     end if;
50                 when RX_Data =>
51                     if clock_count < UART_CLKB - 1 then
52                         clock_count <= clock_count + 1;
53                         rx_state <= RX_Data;
54                     else
55                         clock_count <= 0;
56                         register_byte(bit_index) <= serial_input;
57                         if bit_index < UART_BITS - 1 then
58                             bit_index <= bit_index + 1;
59                             rx_state <= RX_Data;
60                         else
61                             bit_index <= 0;
62                             rx_state <= RX_Stop;
63                         end if;
64                     end if;
65                 when RX_Stop =>
66                     if clock_count < UART_CLKB - 1 then
67                         clock_count <= clock_count + 1;
68                         rx_state <= RX_Stop;
69                     else
70                         wire_stop_bit <= '1';
71                         clock_count <= 0;
72                         rx_state <= RX_Cleanup;
73                     end if;
74                 when RX_Cleanup =>
75                     rx_state <= RX_Idle;
76                     wire_stop_bit <= '0';
77                 when others =>
78                     rx_state <= RX_Idle;
79             end case;
80         end if;
81     end process;
82
83     stop_bit <= wire_stop_bit;
84     data_byte <= register_byte;
85 end architecture;

```

Verification

A testbench is code that applies time stimulus to a design, observes the response, and verifies it with expected behavior. It is basically a software version of a logic analyzer.

Verification is the most important process in digital design. There are substantially more verification engineers than design engineers, they made exorbitant amounts of money, and verification is around 80% of effort in the design process.

There are several ways to write a testbench. We shall explore them as they ramp up in complexity.

Extremely basic testbench

Given a full adder design from page 7, we can write a simple test bench to stimulate the inputs and then report the outputs to the terminal:

```
1  library ieee;
2  use ieee.std_logic_1164.all;
3
4  entity fa_tb is
5  end entity;
6
7  architecture test of fa_tb is
8      signal ain,bin,cin,summation,cout : std_logic := '0';
9      constant DELAY : time := 10 ns;
10     constant CHECK_DELAY : time := 5 ns;
11 begin
12     fa_inst: entity work.fa
13     port map(
14         a    => ain,
15         b    => bin,
16         c    => cin,
17         sum  => summation,
18         carry => cout
19     );
20
21     process
22     begin
23         ain <= '0';
24         bin <= '0';
25         cin <= '0';
26         wait for CHECK_DELAY;
27         report "Sum: " & to_string(summation);
28         report "Carry: " & to_string(cout);
29         wait for DELAY;
30         -- etc etc etc
31
32         wait; --makes process wait forever to end testbench
33     end process;
34 end architecture;
```

Very nice. As can be seen the testbench stimulates the inputs and then reports the inputs.

Just like processes in synthesis the signals do not get assigned until the process suspends. A process with a sensitivity list is like having a "wait on " statement at the end of a process.

The wait statement is when it suspends. That is why we have a CHECK_DELAY in order to accurately report the Sum and Carry values.

Data structures for Testbenches

Writing out repetitive code for a testbench can be annoying. Thankfully using a **record** data structure type, we can automate this process with a loop.

```
1  library ieee;
2  use ieee.std_logic_1164.all;
3
4  entity fa_tb is
5  end entity;
6
7  architecture test of fa_tb is
8      type pattern_type is record
9          a, b, c : std_logic;
10         o, cot : std_logic;
11     end record;
12     type pattern_array is array (natural range <>) of pattern_type;
13     signal ain,bin,cin,summation,cout : std_logic := '0';
14     constant DELAY : time := 10 ns;
15     constant CHECK_DELAY : time := 5 ns;
16
17     constant PATTERNS: pattern_array :=
18         (('0', '0', '0', '0', '0'),
19          ('0', '0', '1', '1', '0'),
20          ('0', '1', '0', '1', '0'),
21          ('0', '1', '1', '0', '1'),
22          ('1', '0', '0', '1', '0'),
23          ('1', '0', '1', '0', '1'),
24          ('1', '1', '0', '0', '1'),
25          ('1', '1', '1', '1', '1'));
26
27     begin
28         fa_inst: entity work.fa
29         port map(
30             a => ain,
31             b => bin,
32             c => cin,
33             sum => summation,
34             carry => cout
35         );
36
37         process
38         begin
39             for i in PATTERNS'range loop
40                 ain <= PATTERNS(i).a;
41                 bin <= PATTERNS(i).b;
42                 cin <= PATTERNS(i).c;
43                 wait for CHECK_DELAY;
44                 assert summation = patterns(i).o
45                     report "Mismatch on sum"
46                     severity error;
47                 assert cout = patterns(i).cot
48                     report "Mismatch on carry"
49                     severity error;
50                 wait for DELAY;
51             end loop;
52             report "End of test!";
53             wait; --makes process wait forever to end testbench
54         end process;
55     end architecture;
```

The keyword `assert` makes it so that if the summation signal does not equal the value of our test vector, it reports an error at the timestamp. If nothing gets reported, everything worked fine!